Proceedings of Korean Data And Information Science Society,
October 28-29, 2005, pp. 103-115

Applications of Maple package in Education of Mathematics for Statistics

Dae-Heung Jang[1)]

Abstract

Mathematical packages have the advantages of symbolic computation and powerful graphics interface in contrast with statistical packages. We can use mathematical packages as a support tool in education of mathematics for statistics.

Keywords : 통계패키지, 수학패키지

1. 머리말

통계학 수업 중 통계수학을 교수할 때 주로 지필수업을 위주로 강의하고 있는 것이 현실이다. 이 때 수학패키지를 수업의 보조수단으로 활용하면 학습효과를 증진시킬 수 있다. 일반에게 널리 알려진 수학패키지로는 MATHEMATICA와 Maple 등이 있다. 이러한 수학패키지는 자료분석의 입장에서는 단점들을 갖고 있으나 수치계산뿐만이 아니라 symbolic computation을 할 수 있고 강력한 그래픽스 기능이 있다는 장점들이 있다. 이러한 장점들은 통계패키지들이 갖고 있지 못한 기능들이다. 통계패키지도 그래픽스 기능이 있으나 통계기법을 중심으로 전개되는 반면 수학패키지는 수학함수를 중심으로 전개된다. 수학패키지를 이용한 통계학교재들로서는 기초통계학, 확률론 및 수리통계학을 중심으로 Abell외 2인(1999), Karian과 Tanis(1999), Hastings(2000), Rose와 Smith(2002), Rafter외 2인(2003) 등이 있다.

2. 수학패키지의 활용

통계수학의 교수내용은 대략 다음과 같다.

1. 집합과 함수
2. 미분법
3. 적분법
4. 무한급수와 함수의 전개
5. 선형대수

1) 부산광역시 남구 대연3동 599-1 부경대학교 수리과학부 통계학전공 교수
E-mail : dhjang@pknu.ac.kr

6. 다변수함수해석
7. 조합론

위의 같은 통계수학 내용의 교수에서 학습효과를 증진시키기 위하여 수학패키지를 수업의 보조수단으로 활용하여 볼 수 있다. 수학패키지로서 Maple을 사용하였다. 통계수학 내용 중 각 단원에 대하여 Khuri(1993)의 책에 나오는 예들을 이용하여 수학패키지를 활용하는 방법에 대하여 다음과 같이 서술할 수 있다. 편의상 각 단원의 내용 중 일부분만을 다루었다.

1. 집합과 함수

함수의 연속은 다음과 같이 엄밀히 정의된다.

(함수의 연속) 임의의 양수 ϵ에 대하여 $|x-a|<\delta$이면 $|f(x)-f(a)|<\epsilon$이 만족되는 $\delta>0$가 존재하는 경우 $x=a$에서 $f(x)$는 연속이라고 한다.

$x=a$에서 함수 $f(x)$가 연속이려면 $\lim_{x \to a} f(x) = f(a)$가 성립되어야 한다.

예 1: 다음과 같이 함수를 정의할 때

$$f(x) = \begin{cases} \cos\left(\dfrac{1}{x}\right), & x \neq 0 \\ 0, & x = 0 \end{cases}$$

$x=0$에서 함수 $f(x)$가 연속인지 확인하기 위하여 Maple에서 다음과 같이 함수를 정의하고 그림을 그리고 $x=0$에서의 좌, 우 극한값을 구하여 볼 수 있다. $x=0$에서 함수 $f(x)$가 불연속임을 알 수 있다.

```
> f3:=piecewise(x=0,0,cos(1/x));
> plot(f3(x),x=-3..3);
> plot(f3(x),x=-0.001..0.001);
> Limit(cos(1/x),x=0,right)=limit(cos(1/x),x=0,right);
> Limit(cos(1/x),x=0,left)=limit(cos(1/x),x=0,left);
```

다음 그림 1(a)와 (b)는 각각 $f(x)$의 그림과 $x=0$ 근방에서의 $f(x)$의 그림을 나타낸다.



(a) (b)

그림 1. $\cos\left(\frac{1}{x}\right)$의 그림

2. 미분법

우리는 함수의 미분을 이용하여 함수의 극한, 증감과 오목/볼록 여부 등을 판정할 수 있다.

예 2: 다음과 같이 함수를 정의할 때

$$f(x) = \begin{cases} x^2 sin\left(\frac{1}{x}\right), & x \neq 0 \\ 0, & x = 0 \end{cases}$$

$x=0$에서 함수 $f(x)$가 연속인지 확인하기 위하여 Maple에서 다음과 같이 함수를 정의하고 그림을 그리고 $x=0$에서의 좌, 우 극한값을 구하여 볼 수 있다. $x=0$에서 함수 $f(x)$가 연속임을 알 수 있다. 그러나 일차도함수 $f'(x)$는 $x=0$에서 불연속임을, 즉 $f(x)$가 미분 불가능함을 Maple을 통하여 확인할 수 있다.

```
> f:=piecewise(x=0,0,x^2*sin(1/x));
> plot(f(x),x=-0.35..0.35);
> plot(f(x),x=-0.005..0.005);
> limit(x^2*sin(1/x),x=0);
> limit(x^2*sin(1/x),x=0,right);
> limit(x^2*sin(1/x),x=0,left);
> Diff(x^2*sin(1/x),x)=diff(x^2*sin(1/x),x);
> diff_f:=x->2*x*sin(1/x)-cos(1/x);
> diff_f(0);
> plot(diff_f(x),x=-5..5);
> plot(diff_f(x),x=-0.005..0.005);
> Limit(2*x*sin(1/x)-cos(1/x),x=0,right)=limit(2*x*sin(1/x)-cos(1/x),x=0,right);
```

```
> Limit(2*x*sin(1/x)-cos(1/x),x=0,left)=limit(2*x*sin(1/x)-cos(1/x),x=0,left);
```

다음 그림 2(a)와 (b)는 각각 $x=0$ 근방에서의 $f(x)$의 그림과 $f'(x)$의 그림을 나타낸다.



(a) (b)

그림 2. $f(x)=x^2 sin\left(\frac{1}{x}\right)$와 $f'(x)$의 그림

예 3: 함수 $f(x)=x^x\,(x \geq 0)$에서 $f(x)$의 $x=0$에서의 극한, 증감과 오목/볼록 여부 등을 판정하기 위하여 아래와 같이 Maple에서 $f'(x)$와 $f''(x)$를 구하고 이를 이용하여 극값(최소값)을 구하면 $e^{-1} \doteq 0.3679$임을 알 수 있고 이 함수는 아래로 볼록함을 알 수 있다.

```
> f2:=x->x^x;
> Limit(x^x,x=0,right)=limit(x^x,x=0,right);
> plot(x^x,x=0..2,y=0..2);
> f2(0);
> Diff(x^x,x)=diff(x^x,x);
> diff_f2:=x->x^x*(ln(x)+1);
> plot(diff_f2(x),x=0..2);
> solve(diff_f2(x)=0,x);evalf(solve(diff_f2(x)=0,x));
> Diff(diff(x^x,x),x)=diff(diff(x^x,x),x);
> plot(x^x*(ln(x)+1)^2+x^x/x,x=0..2,y=0..10);
```

다음 그림 3(a), (b), (c)는 각각 $f(x)$, $f'(x)$, $f''(x)$를 나타내는 그림이다.



(a) (b) (c)

그림 3. $f(x) = x^x$, $f'(x)$, $f''(x)$의 그림

3. 적분법

(정리) 함수 $f(x)$가 $b \ge a$를 만족하는 $[a,b]$에서 리만적분이 가능한 비음함수라 하자. 그리고 $x \ge a$에서 $f(x) \le g(x)$를 만족하는 함수 $g(x)$가 존재한다고 하자. $\int_a^\infty g(x)dx$가 수렴하면 $\int_a^\infty f(x)dx$도 수렴하고 $\int_a^\infty f(x)dx \le \int_a^\infty g(x)dx$가 성립한다.

위의 정리를 이용하여 아래와 같은 이상적분을 고려하여 보자.

예 4: $\frac{|\sin(x)|}{(x+1)^2} \le \frac{1}{(x+1)^2}$을 만족하고 $\int_a^\infty \frac{1}{(x+1)^2}dx = 1$이 됨으로 $\int_a^\infty \frac{\sin(x)}{(x+1)^2}dx$는 절대수렴한다. 또한 수렴하는 데 그 수렴값은 $\frac{1}{2}\pi \sin(1) - C_i(1)\cos(1) - S_i(1)\sin(1) \doteq 0.3434$가 된다. 여기서 $S_i(x)$는 sine integral을, $C_i(x)$는 cosine integral을 의미한다.

다음 그림 4(a)는 $\frac{\sin(x)}{(x+1)^2}$과 $\frac{1}{(x+1)^2}$의 그림을 나타내고 그림 4(b)는 x에 따른 $\int_a^x \frac{\sin(t)}{(t+1)^2}dt$의 값의 변화를 나타내는 그림이다.


(a)



(b)

그림 4. $\dfrac{sin(x)}{(x+1)^2}$ 과 $\dfrac{1}{(x+1)^2}$ 의 그림 및 $\displaystyle\int_a^x \frac{sin(t)}{(t+1)^2}dt$의 그림

위의 작업에 대한 Maple 프로그램 작성은 다음과 같다.

```
> p5:=plot(sin(x)/(x+1)^2,x=0..10,color=red):
> p6:=plot(1/(x+1)^2,x=0..10,color=blue):
> p7:=plot(abs(sin(x))/(x+1)^2,x=0..10,color=green):
> display({p5,p6});
>
Int(sin(x)/(x+1)^2,x=0..infinity)=int(sin(x)/(x+1)^2,x=0..infinity);evalf(int(sin(
x)/(x+1)^2,x=0..infinity));
> Int(1/(x+1)^2,x=0..infinity)=int(1/(x+1)^2,x=0..infinity);
> Int(abs(sin(x))/(x+1)^2,x=0..infinity)=int(abs(sin(x))/(x+1)^2,x=0..infinity);
> int_f:=x->int(sin(t)/(t+1)^2,t=0..x);
> int_f1:=x->int(abs(sin(t))/(t+1)^2,t=0..x);
>
Limit(int(sin(t)/(t+1)^2,t=0..x),x=infinity)=limit(int(sin(t)/(t+1)^2,t=0..x),x=in
finity);
>
Limit(int(sin(t)/(t+1)^2,t=0..x),x=infinity)=evalf(limit(int(sin(t)/(t+1)^2,t=0..x
),x=infinity));
> plot({int_f(x),.3433779610},x=0..100,labels=[x,integral]);
> int_f(x);
> int_f1(x);
```

4. 무한급수와 함수의 전개

무한급수의 수렴, 발산 여부는 적분판정법, 비교판정법, 비례판정법, root판정법 등을 이용하여 판정할 수 있다.

예 5: $\sum_{k=1}^{\infty}\frac{sin(k)}{k}$의 수렴성을 알기 위하여 k에 따른 $a_k=\frac{sin(k)}{k}$과 $\sum_{n=1}^{k}\frac{sin(n)}{n}$의 그림을 그리면 각각 그림 5(a)와 그림 5(b)와 같다. 이 무한급수의 수렴값은 1.0709이다. 적분판정법을 이용하면 $\int_1^{\infty}\frac{sin(x)}{x}dx=\lim_{b\to\infty}\int_1^{b}\frac{sin(x)}{x}dx=\frac{\pi}{2}-S_i(1)\doteq 0.6247$이 되어 이 무한급수는 수렴함을 알 수 있다.



(a) (b)

그림 5. $a_k=\frac{sin(k)}{k}$과 $\sum_{n=1}^{k}\frac{sin(n)}{n}$의 그림

위의 작업에 대한 Maple 프로그램 작성은 다음과 같다.

```
> l16:=[seq([k,sin(k)/k],k=1..1000)]:
> pointplot(l16,axes=boxed,labels=[k,"sin(k)/k"]);
> f16:=n->evalf(sum(sin(k)/k,k=1..n)):
> f16(10000);
> l17:=[seq([n,f16(n)],n=1..1000)]:
> p16_1:=pointplot(l17,axes=boxed):
> p16_2:=plot(1.070868190,x=1..1000,axes=boxed):
> display({p16_1,p16_2},labels=[k,sum]);
```

```
> int(sin(x)/x,x=1..infinity);
> evalf(int(sin(x)/x,x=1..infinity));
```

5. 선형대수

우리는 Maple을 이용하여 행렬에 대한 규칙들을 수치적으로 뿐만이 아니라 심볼을 이용하여, 즉 대수적으로 다양하게 경험하여 볼 수 있다.

예 6: Maple을 이용하여 각 원소가 심볼로 주어지는 3×3 행렬 A의 행렬식을 구하고 역행렬 A^{-1}을 구한 후 $A^{-1}=\frac{1}{det(A)}adj(A)$가 성립함을 확인할 수 있다.

```
> A:=matrix(3,3,[a11,a12,a13,a21,a22,a33,a31,a32,a33]);
> trace(A);
> rank(A);
> det(A);
> inverse(A);
The Minor
> for i from 1 to 3 do
> for j from 1 to 3 do
>
M(i,j):=minor(A,i,j);print("minor");print("row=",i);print("col=",j);print(M(i,j));
> end do
> end do;
The Adjoint
> adjoint(A);
inverse(A) = adj(A)/det(A)
> scalarmul(adjoint(A),1/det(A));
```

예 7: Maple을 이용하여 각 원소가 심볼로 주어지는 2×2 행렬 $A3$의 특성방정식, 고유값과 고유벡터, 고유값과 행렬식 또는 트레이스와의 관계를 확인하여 본다.

```
Eigenvalues & Eigenvectors of a Matrix
> A3:=matrix(2,2,[a,b,c,d]);
> charmat(A3,lambda);
> det(A3);
> trace(A3);
> inverse(A3);
> eigenvals(A3);
> solve(det(charmat(A3,lambda))=0,lambda);
> lambda1:=1/2*a+1/2*d+1/2*sqrt(a^2-2*a*d+d^2+4*b*c);
```

```
> lambda2:=1/2*a+1/2*d-1/2*sqrt(a^2-2*a*d+d^2+4*b*c);
> lambda1+lambda2;trace(A3);
> simplify((lambda1*lambda2)-det(A3));
> eigenvects(A3);
```

예 8: Maple을 이용하여 spectral decomposition theorem을 확인하여 볼 수 있다.

The Diagonalization of a Matrix

```
> A4:=matrix(3,3,[1,0,-2,0,0,0,-2,0,4]);
> eigenvals(A4);
> L:=diag(0,0,5);
> eigenvects(A4);
> p1:=scalarmul(vector([2, 0, 1]),1/sqrt(5));
> p2:=scalarmul(vector([0, 1, 0]),1/sqrt(5));
> p3:=scalarmul(vector([1, 0, -2]),1/sqrt(5));
> P:=augment(p1,p2,p3);       > evalm(P&*L&*transpose(P));
```

예 9: Maple을 이용하여 3×3 대칭행렬의 이차형식과 고유값 사이의 관계를 알아본다.

Quadratic Forms

```
> A1:=matrix(3,3,[2,3,-1,0,1,2,2,4,1]);
> eigenvals(A1);
> v1:=vector([x,y,z]);
> simplify(evalm(transpose(v1)&*A1&*v1));
> f1:=(x,y,z)->evalm(transpose(v1)&*A1&*v1);
>
p1_1:=implicitplot3d(f1(x,y,z)=-10,x=-2..2,y=-2..2,z=-2..2,axes=boxed,color=yellow
,labels=[x1,x2,x3]):
>
p1_2:=implicitplot3d(f1(x,y,z)=-5,x=-2..2,y=-2..2,z=-2..2,axes=boxed,color=cyan):
>
p1_3:=implicitplot3d(f1(x,y,z)=0,x=-2..2,y=-2..2,z=-2..2,axes=boxed,color=magenta)
:
>
p1_4:=implicitplot3d(f1(x,y,z)=5,x=-2..2,y=-2..2,z=-2..2,axes=boxed,color=green):
>
p1_5:=implicitplot3d(f1(x,y,z)=10,x=-2..2,y=-2..2,z=-2..2,axes=boxed,color=violet)
:
> display(p1_1,p1_2,p1_3,p1_4,p1_5);print("yellow: -10, cyan: -5, magenta: 0,
green: 5, violet: 10");
```

6. 다변수함수해석

1장부터 4장까지의 내용을 다변수함수로 확장하여 다변수함수의 극한, 연속, 미분, 최대/최소, 적분 등을 고려하여 볼 수 있다.

예 10: 다변수함수 $f(x_1,x_2)=\dfrac{x_2(x_1^2+x_2^2)}{x_2^2+(x_1^2+x_2^2)^2}$는 원점 (0,0)를 제외하고 R^2 상에서 정의가 된다. (0,0)에서의 극한값이 존재하는가를 조사하기 위하여 이 함수를 Maple을 이용하여 그려보고 $x_1=0$을 만족하는 곡선과 $x_2=0$를 만족하는 직선을 그려 (0,0)에서의 극한값을 구하여 보면 0가 됨을 알 수 있다. 그러나 원점을 지나는 원 $x_2=x_1^2+x_2^2$을 따라가면 (0,0)에서의 $f(x_1,x_2)$의 값은 0.5가 되어 결국 (0,0)에서의 극한값이 존재하지 않음을 알 수 있다. 위의 작업에 대한 Maple 프로그램 작성은 다음과 같다.

```
> f3:=(x,y)->y*(x^2+y^2)/(y^2+(x^2+y^2)^2);
> plot3d(f3(x,y),x=-5..5,y=-5..5,axes=boxed,labels=[x1,x2,f]);
> contourplot3d(f3(x,y),x=-5..5,y=-5..5,axes=boxed,labels=[x1,x2,f]);
> contourplot(f3(x,y),x=-5..5,y=-5..5,axes=boxed,labels=[x1,x2]);
> f(0,0);
> f4:=y->y^3/(y^2+y^4);
> plot(f4(y),y=-5..5,labels=[x2,f]);
> f5:=x->0;
> plot(f5(x),x=-5..5,labels=[x1,f],color=red,axes=boxed);
> f6:=x->5*x*(x^2+(5*x)^2)/((5*x)^2+(x^2+(5*x)^2)^2);
> plot(f6(x),x=-5..5,labels=[x1,f],color=red);
> p4:=plot3d(f3(x,y),x=-5..5,y=-5..5,axes=boxed,labels=[x1,x2,f],color=yellow):
> p5:=plot3d(f4(y),x=-0.1..0.1,y=-5..5,axes=boxed,labels=[x1,x2,f],color=cyan):
> p6:=plot3d(f5(x),x=-5..5,y=-0.1..0.1,axes=boxed,labels=[x1,x2,f],color=magenta):
> p7:=plot3d(f6(x),x=-5..5,y=-5..5,axes=boxed,labels=[x1,x2,f],color=black):
> display({p4,p5,p6});
> implicitplot(x^2+y^2-y=0,x=-1..1,y=-1..1,labels=[x1,x2]);
```

그림 6은 $f(x_1,x_2)$ 곡면 위에 $x_1=0$을 만족하는 곡선과 $x_2=0$를 만족하는 직선을 그려 넣은 그림이다.



그림 6. $f(x_1, x_2)$ 곡면 위에 $x_1 = 0$을 만족하는 곡선과 $x_2 = 0$를 만족하는 직선을 그려 넣은 그림

7. 조합론

순열과 조합은 이산적 확률분포와 연속적 분포에서 쓰인다.

예 11: Maple을 이용하여 파스칼의 삼각형, 이항계수의 확장인 다항계수를 계산하여 본다.

Pascal's Triangle

```
> for i from 1 to 10 do
> expand((p+q)^i);
> end do;
> for j from 0 to 10 do
> binomial(10,j)
> end do;
> plot3d(binomial(floor(n),floor(r)),n=1..10,r=1..n,axes=boxed);
> binomial(10,4);
> binomial(9,3)+binomial(9,4);
> binomial(n,r)-binomial(n-1,r-1)-binomial(n-1,r);
```

Multinomial Coefficient

```
> multinomial(8, 2, 3, 3); 8!/(2!*3!*3!);
```

예 12: Maple을 이용하여 이항분포와 포아송분포에서 평균, 분산, 적률발생함수, 정규근사에 대하여 확인한다.

Binomial Distribution

```
> f1:=x->binomial(n,x)*p^x*(1-p)^(n-x);   # pmf of binomial distribution
> sum(f1(x),x=0..n);
> simplify(sum(x*f1(x),x=0..n));simplify(sum((x-n*p)^2*f1(x),x=0..n));
                                        # mean and variance of binomial distribution
> simplify(sum(exp(t*x)*f1(x),x=0..n));   # mgf of binomial distribution
> m1:=t->(1-p+p*exp(t))^n;        > dm1:=t->diff(m1(t),t);
> EX1:=simplify(subs(t=0,dm1(t)));       > ddm1:=t->diff(dm1(t),t);
> EXX1:=simplify(subs(t=0,ddm1(t)));     > EXX1-EX1^2;
> F1:=x->sum(f1(t),t=0..x);
> n:=10;p:=1/2;for i from 0 to 10 do print (i,f1(i))od;
> f1:=x->binomial(n,floor(x))*p^floor(x)*(1-p)^(n-floor(x));
> plot(F1(x),x=0..11,y=0..1);       > p3:=plot(F1(x),x=0..11,y=0..1):
> n:=10;p:=1/6;       > plot(F1(x),x=0..11,y=0..1,color=blue);
> p4:=plot(F1(x),x=0..11,y=0..1,color=blue):      > display({p3,p4});
> n:=50;p:=0.1;       > p5:=plot(F1(x),x=-5..15,y=0..1):
>
F4:=x->int(1/sqrt(2*Pi*n*p*(1-p))*exp(-1/(2*n*p*(1-p))*(t-n*p)^2),t=-infinity..x);
> p6:=plot(F4(x),x=-5..15,y=0..1,color=blue):
> display({p5,p6});   # normal approximation to binomial
```

Poisson Distribution

```
> f2:=x->exp(-mu)*mu^x/x!;   # pmf of Poisson Distribution
> E1:=sum(x*f2(x), x=0..infinity);
> simplify(E1);   # E(X) of Poisson Distribution
> E2:=sum(x*(x-1)*f2(x), x=0..infinity);
> simplify(E2);       > V:=E2+E1-E1^2;
> simplify(V);   # Var(X) of Poisson Distribution
> simplify(sum(exp(t*x)*f2(x),x=0..infinity));   # mgf of Poisson distribution
> m2:=t->exp(mu*(-1+exp(t)));       > dm2:=t->diff(m2(t),t);
> EX2:=simplify(subs(t=0,dm2(t)));    > ddm2:=t->diff(dm2(t),t);
> EXX2:=simplify(subs(t=0,ddm2(t)));    > EXX2-EX2^2;
> mu:=1;for i from 0 to 10 do print (i,evalf(f2(i)))od;
> F2:=x->sum(f2(t),t=0..x);     > f2:=x->exp(-mu)*mu^floor(x)/floor(x)!;
> plot(F2(x),x=0..11,y=0..1);
> mu:=5;     > p1:=plot(F2(x),x=-5..15,y=0..1):
> F3:=x->int(1/sqrt(2*Pi*mu)*exp(-1/(2*mu)*(t-mu)^2),t=-infinity..x);
> p2:=plot(F3(x),x=-5..15,y=0..1,color=blue):
> display({p1,p2});   # normal approximation to Poisson
```

참고문헌

1. Abell, M. L., Braselton, J. P., and Rafter, J. A.(1999). *Statistics with Mathematica*, Academic Press.
2. Hastings, K. J.(2000). *Introduction to Probability with Mathematica*, Lewis Publishers, Inc.
3. Karian, Z. A. and Tanis, E. A.(1999). *Probability and Statistics: Exploration with Maple*, Prentice Hall.
4. Khuri, A.I. (1993). *Advanced Calculus with Applications in Statistics,* John Wiley & Son.
5. Rafter, J. A., Braselton, J. P., and Abell, M. L.(2003). *Statistics with Maple*, Academic Press.
6. Rose, C. and Smith, M. D.(2002). *Mathematical Statistics with Mathematica*, Springer Verlag.